MATRIX PROJECT PARTIAL EXAMPLE

```
> with(LinearAlgebra):
```

Let us try to learn everything we can about the matrix A below. One thing we can easily note by looking at the matrix is that it is symmetric. The other thing we might not is that the matrix has cross-diagonals that are constant . To be clear about this we note that $A_{3,1} = A_{2,2} = A_{1,3} = \frac{1}{4}$ and that $A_{4,1} = A_{3,2} = A_{2,3} = A_{1,4} = \frac{1}{8}$. It is unclear what the number $\frac{64}{85}$ does for this matrix. We will hopefully find out as we go along. One other thing we might note is that if we multiply the first column by $\frac{1}{2}$ we will get the second column. If we multiply the first column by $\frac{1}{4}$ and then by $\frac{1}{8}$ we will get the third and fourth columns respectively. Thus the second, third, and fourth columns are scalar multiples of the first. This says that the rank(A) =1 since the first column will span Col(A). Note also that this observation about the columns is true for the rows also since the matrix is symmetric.

$$A := \begin{bmatrix} \frac{64}{85} & \frac{32}{85} & \frac{16}{85} & \frac{8}{85} \\ \frac{32}{85} & \frac{16}{85} & \frac{8}{85} & \frac{4}{85} \\ \frac{16}{85} & \frac{8}{85} & \frac{4}{85} & \frac{2}{85} \\ \frac{8}{85} & \frac{4}{85} & \frac{2}{85} & \frac{1}{85} \end{bmatrix}$$

As a way of getting started with an analysis of A we could consider $\frac{85\,A}{64}$ and get

$> M := \frac{85}{64} A$

$$M := \begin{bmatrix} 1 & \frac{1}{2} & \frac{1}{4} & \frac{1}{8} \\ \frac{1}{2} & \frac{1}{4} & \frac{1}{8} & \frac{1}{16} \\ \frac{1}{4} & \frac{1}{8} & \frac{1}{16} & \frac{1}{32} \\ \frac{1}{8} & \frac{1}{16} & \frac{1}{32} & \frac{1}{64} \end{bmatrix} \qquad (1)$$

Now, if we let

•

$$v := \left\langle 1, \frac{1}{2}, \frac{1}{4}, \frac{1}{8} \right\rangle \tag{2}$$

```
> v:=<1,1/2,1/4,1/8>;
```

$$v := \begin{bmatrix} 1 \\ \frac{1}{2} \\ \frac{1}{4} \\ \frac{1}{8} \end{bmatrix} \tag{3}$$

then the matrix A = $\frac{64}{85}\ v\, v^T$. We will examine later in this exposition what this way of constructing a matrix will lead to.

```
> A:=64/85*v.Transpose(v);
```

$$A := \begin{bmatrix} \frac{64}{85} & \frac{32}{85} & \frac{16}{85} & \frac{8}{85} \\ \frac{32}{85} & \frac{16}{85} & \frac{8}{85} & \frac{4}{85} \\ \frac{16}{85} & \frac{8}{85} & \frac{4}{85} & \frac{2}{85} \\ \frac{8}{85} & \frac{4}{85} & \frac{2}{85} & \frac{1}{85} \end{bmatrix} \tag{4}$$

We note that since A has rank 1 it will not be invertible and hence its determinant will be 0. We confirm this with the following Maple command.

```
> Determinant(A);
```

$$0 \tag{5}$$

Since A is not invertible it will have a nontrivial nullspace. The nullspace is spanned by the three vectors given below. We note that this also confirms that the rank is 1, since rank + nullity = number of columns = 4.

```
> NullSpace(A);
```

$$\left\{ \begin{bmatrix} -\frac{1}{4} \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{8} \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -\frac{1}{2} \\ 1 \\ 0 \\ 0 \end{bmatrix} \right\} \tag{6}$$

```
> Rank(A);
```

$$1 \tag{7}$$

We now will find both the charateristic polynomial of A and also all eigenvalues along with their associated eigenvectors.

```
> p:=lambda->CharacteristicPolynomial(A,lambda);
```

$$p := \lambda \rightarrow LinearAlgebra\text{:-}CharacteristicPolynomial(A, \lambda) \tag{8}$$

```
> p(lambda);
```

$$\lambda^4 - \lambda^3 \tag{9}$$

```
> solve(p(lambda)=0,lambda);
```

$$0, 0, 0, 1 \tag{10}$$

We see from this that 0 is an eigenvalue. We already knew this since null(A) was nontrivial. Any eigenvector associated with the eigenvalue 0 is an element of null(A). In fact any nonzero element of null(A) is an eigenvector with eigenvalue = 0. Now we discover the eigenvectors for the eigenvalue 1. We can also get eigenvectors and eigenvalues using the Eigenvectors command.

```
> eigs:=Eigenvectors(A,output=list);
```

$$eigs := \left[\left[0, 3, \left\{ \begin{bmatrix} -\frac{1}{2} \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{4} \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{1}{8} \\ 0 \\ 0 \\ 1 \end{bmatrix} \right\} \right], \left[1, 1, \left\{ \begin{bmatrix} 8 \\ 4 \\ 2 \\ 1 \end{bmatrix} \right\} \right] \right] \tag{11}$$

From this we confirm that 0 is an eigenvalue of A with algebraic and geometric multiplicity 3. We also see that 1 is an eigenvalue of A with both algebraic and geometric multiplicity equal to 1. Since all algebraic and geometric multiplicities are equal, the matrix A will be diagonalizable. We can do that using:

```
> S:=<eigs[1][3][1]|eigs[1][3][2]|eigs[1][3][3]|eigs[2][3][1]>;
```

$$S := \begin{bmatrix} -\frac{1}{8} & -\frac{1}{4} & -\frac{1}{2} & 8 \\ 0 & 0 & 1 & 4 \\ 0 & 1 & 0 & 2 \\ 1 & 0 & 0 & 1 \end{bmatrix} \tag{12}$$

```
> Di:=MatrixInverse(S).A.S;
```

$$Di := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{13}$$

```
> A-A^2;
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{14}$$

This command tells us that A = A*A so in fact any postive integer power of A will be A again: (A*A*A = A*(A*A) = A*A = A etc.).

```
> ReducedRowEchelonForm(A);
```

$$\begin{bmatrix} 1 & \frac{1}{2} & \frac{1}{4} & \frac{1}{8} \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \qquad (15)$$

This is the reduced, row echelon form of A which we should have guessed since rank(A) = 1, and hence the first row of A would serve as a basis for the rowspace(A).

We examine several of the norms for a matrix. For example, the spectral norm is

```
> MatrixNorm(A,2);
```

$$1 \qquad (16)$$

The 1 norm is the maximum of the sum of the absolute values of the column entries. This turns out to be:

```
> MatrixNorm(A,1);
```

$$\frac{24}{17} \qquad (17)$$

Since the matrix is symmetric the 1 norm of A and the ∞ norm of A should be the same. The following shows that they are.

```
> MatrixNorm(A,infinity);
```

$$\frac{24}{17} \qquad (18)$$

The Froebenius norm of A is the square root of the sum of the squares of the entries. For that we get:

```
> MatrixNorm(A,Frobenius);
```

$$1 \qquad (19)$$

```
> Trace(A);
```

$$1 \qquad (20)$$

Now let's examine what happens when we construct any matrix the same way I did for A in Maple. So let w be the 4x1 matrix with arbitrary entries a,b,c, and d. So

```
> u:=<a,b,c,d>;
```

$$u := \begin{bmatrix} a \\ b \\ c \\ d \end{bmatrix} \qquad (21)$$

Now C is our matrix of interest.

```
> C:=u.Transpose(u);
```

(22)

$$C := \begin{bmatrix} a^2 & a\,b & a\,c & a\,d \\ a\,b & b^2 & b\,c & b\,d \\ a\,c & b\,c & c^2 & c\,d \\ a\,d & b\,d & c\,d & d^2 \end{bmatrix} \tag{22}$$

We note that C has rank 1 just like A and hence will not be invertible leading us to det(C) = 0:

```
> Determinant(C);
```

$$0 \tag{23}$$

```
> Rank(C);
```

$$1 \tag{24}$$

```
> NullSpace(C);
```

$$\left\{ \begin{bmatrix} -\frac{b}{a} \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{d}{a} \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -\frac{c}{a} \\ 0 \\ 1 \\ 0 \end{bmatrix} \right\} \tag{25}$$

We note that the output above makes sense as long as a≠0. If a=0 then the first column of C will be all zeros so that [1,0,0,0] will be in N(C). More generally, by multiplying each vector above by the scalar "a", the nullspace will be spanned by [-d,0,0,a], [-c,0,a,0], and [-b,a,0,0] .

```
> q:=lambda->CharacteristicPolynomial(C,lambda);
```

$$q := \lambda \to LinearAlgebra:\text{-}CharacteristicPolynomial(C, \lambda) \tag{26}$$

```
> q(lambda);
```

$$\lambda^4 + \left(-d^2 - c^2 - b^2 - a^2\right) \lambda^3 \tag{27}$$

```
> collect(%,lambda);
```

$$\lambda^4 + \left(-d^2 - c^2 - b^2 - a^2\right) \lambda^3 \tag{28}$$

Note that there are really only two terms here, one which involves the fourth power of w3 and the other which involves the third power of w3. We note that the matrix A had $a^2 + b^2 + c^2 + d^2 = 1$. In fact this was the reason for the scalar 64/85 which appears in front of the matrix defining A.

```
> solve(q(lambda)=0,lambda);
```

$$0, 0, 0, d^2 + c^2 + b^2 + a^2 \tag{29}$$

```
> eig:=Eigenvectors(C,output=list);
```

$$eig := \left[\left[d^2 + c^2 + b^2 + a^2, 1, \left\{ \begin{bmatrix} \frac{a}{d} \\ \frac{b}{d} \\ \frac{c}{d} \\ 1 \end{bmatrix} \right\} \right], \left[0, 3, \left\{ \begin{bmatrix} -\frac{d}{a} \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -\frac{c}{a} \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -\frac{b}{a} \\ 1 \\ 0 \\ 0 \end{bmatrix} \right\} \right] \right] \tag{30}$$

This confirms that if a ≠ 0 then any matrix of this form will be similar to the particular matrix originally given. They will be diagonalizable since all algebraic multiplicities will be equal to the geometric multiplicities. The diagonalizing matrix will be:

```
> T:=<<-d,0,0,a>|<-b,a,0,0>|<-c,0,a,0>|<a,b,c,d>>;
```

$$T := \begin{bmatrix} -d & -b & -c & a \\ 0 & a & 0 & b \\ 0 & 0 & a & c \\ a & 0 & 0 & d \end{bmatrix} \quad (31)$$

```
> di:=MatrixInverse(T).C.T:
```

I suppressed the output of this command because it was quite lengthy since it was unsimplified.

```
> map(simplify,di);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & d^2 + c^2 + b^2 + a^2 \end{bmatrix} \quad (32)$$

The following sequence of commands tries to determine what it will take for the square of the matrix to be itself. The colon at the end of the command suppresses output. I have only omitted it because it is lengthy and not valuable to examine.

```
> F:=C^2-C:
> s:=seq(seq(F[i,j]=0,i=1..4),j=1..4):
> solve({s},{a,b,c,d});
```

$$\{d=0, a=0, b=0, c=0\}, \{d=0, a=0, b=0, c=0\}, \{d=d, c=\mathrm{RootOf}(\_Z^2 - 1 + d^2), a=0, b=0\}, \{d=d, c=c, a=0, b=\mathrm{RootOf}(\_Z^2 - 1 + c^2 + d^2)\}, \{d=0, a=1, b=0, c=0\}, \{d=0, a=-1, b=0, c=0\}, \{b=b, d=d, c=c, a=\mathrm{RootOf}(\_Z^2 - 1 + b^2 + c^2 + d^2)\} \quad (33)$$

Note that many solutions involve some or all of the entries being zero. However, in looking at the last solution, we can see that for C^2-C to be the zero matrix we must have the sums of the squares of a,b,c, d to be 1. This illustrates one reason why 64/85 was used in defining the matrix A.

Just to check this result, we try:

```
> G:=map(factor,F);
```

$$G := \Big[[a^2 (a^2 + b^2 + c^2 + d^2 - 1), a\, b\, (a^2 + b^2 + c^2 + d^2 - 1), a\, c\, (a^2 + b^2 + c^2 + d^2 - 1), \quad (34)$$
$$a\, d\, (a^2 + b^2 + c^2 + d^2 - 1)], [a\, b\, (a^2 + b^2 + c^2 + d^2 - 1), b^2 (a^2 + b^2 + c^2 + d^2 - 1),$$
$$b\, c\, (a^2 + b^2 + c^2 + d^2 - 1), b\, d\, (a^2 + b^2 + c^2 + d^2 - 1)], [a\, c\, (a^2 + b^2 + c^2 + d^2 - 1),$$
$$b\, c\, (a^2 + b^2 + c^2 + d^2 - 1), c^2 (a^2 + b^2 + c^2 + d^2 - 1), c\, d\, (a^2 + b^2 + c^2 + d^2 - 1)], [$$
$$a\, d\, (a^2 + b^2 + c^2 + d^2 - 1), b\, d\, (a^2 + b^2 + c^2 + d^2 - 1), c\, d\, (a^2 + b^2 + c^2 + d^2 - 1),$$

$$d^2\left(a^2+b^2+c^2+d^2-1\right)]$$

```
> H:=matrix(4,4);
> for j from 1 to 4 do for i from 1 to 4 do H[i,j]:=subs(a^2=1-(b^2+
  c^2+d^2),G[i,j]) od od;
> evalm(H);
```

$$H := array\left(1..4, 1..4, [\,]\right) \tag{35}$$

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

We can see from G that if $a^2+b^2+c^2+d^2=1$ then G will be 0. This is substantiated when we substitute $a^2=1-\left(b^2+c^2+d^2\right)$ into each entry of G. We see that C*C=C if and only if either all of a,b,c,d = 0 or the sum of the squares of a,b,c,d is 1.

```
> Trace(C);
```

$$d^2+c^2+b^2+a^2 \tag{36}$$

```
> assume(a,real);assume(b,real);assume(c,real);assume(d,real);
> Norm(C,2);
```

$$a^2+b^2+c^2+d^2 \tag{37}$$

From this we see that the spectral norm of the matrix will be 1 if and only if the sum of the squares of a, b,c,d is 1.

Also, in general, we see that if $A=uu^T$ where $u^T\,u=1$, then A is the rank 1 matrix that is the orthogonal projection onto the Span(u).